A Generating Equation for Integrable Charges

Till Bargheer, Niklas Beisert, Florian Loebbert

Max–Planck–Institut für Gravitationsphysik
Albert–Einstein–Institut
Potsdam–Golm
Germany

`Mathematica` Summer School on Theoretical Physics
June 18, 2009

Framework for arXiv:{0807.5081,0902.0956}

Integrable Spin Chains I

Consider a spin chain model, i.e. a tensor product

$$\ldots \otimes \mathbb{V} \otimes \mathbb{V} \otimes \mathbb{V} \otimes \mathbb{V} \otimes \mathbb{V} \otimes \ldots$$

of vector spaces (generalized spins), all transforming in a common representation of a symmetry algebra $\mathfrak{g}$.
Focus on models with *local* and *homogeneous* interactions/charges:

$$\mathcal{L}_k = \sum_a \mathcal{L}_k(a) = \sum_a$$



This type of spin chain finds application in the computation of anomalous dimensions of local gauge invariant operators of $\mathcal{N} = 4$ SYM. [Minahan Zarembo][Beisert hep-th/0407277]
Gauge theory spin chains are integrable, i.e. they feature an infinite set of commuting charges

$$\mathcal{Q}_r = \sum_k \mathcal{L}_k\,, \quad r = 1, \ldots, \infty\,, \qquad [\mathcal{Q}_r, \mathcal{Q}_s] = 0\,, \qquad \mathcal{Q}_2 = \mathcal{H}\,.$$

Integrable Spin Chains II

The charges of gauge theory spin chains feature a perturbative *range expansion*

$\mathcal{H} = \mathcal{Q}_2 =$  $+ \lambda$  $+ \lambda^2$  $+ \lambda^3 \ldots ,$

where λ is the coupling constant.

- As an example, consider commuting charges on a spin chain with $\mathfrak{gl}(n)$ symmetry. In this case, all symmetry invariant operators $\mathcal{L}_k$ are *permutations*. These are the building blocks of the charges.
- Permutations $\pi \in S_n$ are represented in `Mathematica` as

  `Perm`$[\pi(1), \ldots, \pi(n)]$.

  For example, `Perm[3,4,1,2]` maps $\{1, 2, 3, 4\}$ to $\{3, 4, 1, 2\}$.

Goal: Understand long-range integrable spin chains better!

Deforming Short-Range to Long-Range Chains

Integrable charges can be computed by brute force. *(cf. talk by Florian)* [Beisert Klose]

Different approach: Deform a given set of (short-range, $\lambda = 0$) integrable charges $\mathcal{Q}_r$ through a generating equation [Bargheer Beisert Loebbert]

$$\frac{\mathrm{d}}{\mathrm{d}\lambda}\,\mathcal{Q}_r(\lambda) = i\big[\mathcal{X}(\lambda), \mathcal{Q}_r(\lambda)\big],$$

where $\lambda \approx 0$ is the deformation parameter.

- The form of the generating equation guarantees that the algebra obeyed by the charges is invariant under the deformation. By the Jacobi identity

  $$\frac{\mathrm{d}}{\mathrm{d}\lambda}\,\big[\mathcal{Q}_r(\lambda), \mathcal{Q}_s(\lambda)\big] = i\big[\mathcal{X}(\lambda), [\mathcal{Q}_r(\lambda), \mathcal{Q}_s(\lambda)]\big].$$

  Therefore the structure constants are λ-independent,

$$\big[\mathcal{Q}_r(\lambda), \mathcal{Q}_s(\lambda)\big] = f_{rst}\,\mathcal{Q}_t(\lambda) \quad \xRightarrow{\mathrm{d}/\mathrm{d}\lambda} \quad \frac{\mathrm{d}}{\mathrm{d}\lambda}\,f_{rst} = 0\,.$$

- In particular, if the initial charges commute, $f_{rst} = 0$, also the deformed charges $\mathcal{Q}_r(\lambda)$ commute.
  $\Rightarrow$ *The deformation preserves integrability.*

Deformation Operators $\mathcal{X} = ?$

Generating equation:

$$\frac{\mathrm{d}}{\mathrm{d}\lambda}\,\mathcal{Q}_r(\lambda) = i\big[\mathcal{X}(\lambda), \mathcal{Q}_r(\lambda)\big],$$

What are the required properties for the deformation operator $\mathcal{X}$?

- The commutator between $\mathcal{X}(\lambda)$ and the charges $\mathcal{Q}_r(\lambda)$ has to be well-defined.
- The deformed charges $\mathcal{Q}_r(\lambda)$ should again be local and homogeneous as required by gauge theory.

With suitable operators $\mathcal{X}$, long-range integrable spin chains can be constructed.

What are suitable deformation operators $\mathcal{X}(\lambda)$?

$\mathcal{X}$ = Boost Operators

- One suitable type of operator: Boost operators

$$\mathcal{L}_k = \sum_a \mathcal{L}_k(a) \qquad \Longrightarrow \qquad \mathcal{B}[\mathcal{L}_k] := \sum_a a\, \mathcal{L}_k(a)\,.$$

- In general, the commutator of a homogeneous charge operator $\mathcal{Q}_r$ with a boost $\mathcal{B}[\mathcal{L}_k]$ again yields a boost. However, boosts of the charges $\mathcal{B}[\mathcal{Q}_k]$

$$\frac{\mathrm{d}}{\mathrm{d}\lambda}\, \mathcal{Q}_r(\lambda) = i\big[\mathcal{B}[\mathcal{Q}_k(\lambda)], \mathcal{Q}_r(\lambda)\big],$$

yield again homogeneous operators $\mathcal{Q}_r(\lambda)$ as required.
This is due to the fact that the charges $\mathcal{Q}_r$ commute:



Boost Commutator: Implementation 1/3

Boosts $\mathcal{B}[\pi]$ on the $\mathfrak{gl}(n)$ chain are represented in Mathematica as

$$\texttt{PermB}[\pi(1),\ldots,\pi(n)]\,, \qquad \text{e.g. } \texttt{PermB[4,5,2,1,3]}\,.$$

Boosts of charges can be obtained by

```
B[r_] := Q[r] /.  Perm -> PermB.
```

In order to compute deformed charges, need a Mathematica implementation of the commutator between boosts and permutation operators.

Overview of the commutator method in Mathematica:

```
CommutePermB[X_, Y_] :=
 X /. {X1_PermB :> (Y /. Y1_Perm :> CommutePermB12[X1, Y1])}

CommutePermB12[X_PermB, Y_Perm] := Plus[
  CommutePerm[X /. PermB -> Perm, Y] /. Perm -> PermB,
  CommutePermB12Hom[X, Y] ]

CommutePermB12Hom[X_PermB, Y_Perm] := Sum[
   k (+CombinePerm12[X /. PermB -> Perm, Y, Length[X] + k]
      -CombinePerm12[Y, X /. PermB -> Perm, Length[Y] - k]),
   {k, 1, Length[Y] - 1}]
```

Boost Commutator: Implementation 2/3

First, make the commutator distributive:

```
CommutePermB[X_, Y_] :=
 X /. {X1_PermB :> (Y /. Y1_Perm :> CommutePermB12[X1, Y1])}
```

```
CommutePermB12[X_PermB, Y_Perm] := Plus[
  CommutePerm[X /. PermB -> Perm, Y] /. Perm -> PermB,
  CommutePermB12Hom[X, Y] ]
```

Then, each pair of boosted and unboosted permutation yields boosts (center line) and homogeneous terms (last line),

$$\left[\mathcal{B}[\pi_1], \pi_2\right] = \mathcal{B}\left[[\pi_1, \pi_2]\right] + \text{homogeneous}\,.$$



If π_1 and π_2 commute, the boost part (center line in the box above) vanishes.

Boost Commutator: Implementation 3/3

The homogeneous part of the commutator is implemented as

```
CommutePermB12Hom[X_PermB, Y_Perm] := Sum[
   k (+CombinePerm12[X /. PermB -> Perm, Y, Length[X] + k]
      -CombinePerm12[Y, X /. PermB -> Perm, Length[Y] - k]),
   {k, 1, Length[Y] - 1}]
```

`CommutePerm12[P1,P2,k]` *(cf. Florians talk)* computes the product of two overlapping permutations `P1` and `P2`, where the overlap is specified by `k`.

Example:

`CommutePermB12Hom[PermB[2,1,3],Perm[3,2,4,1]]`

$$
\begin{aligned}
&= +1\cdot \begin{array}{r} \texttt{Perm[2,1,3]} \\ \texttt{Perm[3,2,4,1]}\phantom{\texttt{xx}} \end{array} + 2\cdot \begin{array}{r} \texttt{Perm[2,1,3]} \\ \texttt{Perm[3,2,4,1]}\phantom{\texttt{x}} \end{array} + 3\cdot \begin{array}{r} \texttt{Perm[2,1,3]} \\ \texttt{Perm[3,2,4,1]} \end{array} \\
&\quad -1\cdot \begin{array}{r} \texttt{Perm[3,2,4,1]} \\ \texttt{Perm[2,1,3]} \end{array} - 2\cdot \begin{array}{r} \texttt{Perm[3,2,4,1]}\phantom{\texttt{x}} \\ \texttt{Perm[2,1,3]} \end{array} - 3\cdot \begin{array}{r} \texttt{Perm[3,2,4,1]}\phantom{\texttt{xx}} \\ \texttt{Perm[2,1,3]} \end{array}
\end{aligned}
$$

$\mathrm{k} = 1$ $\qquad$ $\mathrm{k} = 2$ $\qquad$ $\mathrm{k} = 3$

Boundary Identifications: Spectator Legs

On an infinite or periodic chain, we can identify terms whose action differs only at chain boundaries. This is necessary for verifying whether deformed charges commute.

E.g. recall that for local operators: $\mathbin{\vdots}\mathcal{L}_k = \mathcal{L}_k$. *(cf. talk by Florian)*

This implies for boosted operators

$$\mathcal{B}[\mathbin{\vdots}\mathcal{L}_k] = \mathcal{B}[\mathcal{L}_k] - \mathcal{L}_k \,.$$




```
PermB[1, X__] = PermB[X-1] - Perm[X-1]
```

Method that implements the identification in `Mathematica`:

```
IdentifyBoundaryTermsBoostLeft[P_] := P //. {
    PermB[X__ /; First[{X}] == 1 && {X} != {1}] :>
      PermB @@ (Drop[{X}, 1] - 1) - Perm @@ (Drop[{X}, 1] - 1)
    }
```

$\mathcal{X}$ = Bilocal Operators

Another suitable type of deformation operators are *bilocal operators*:

$$\mathcal{L}_k = \sum_a \mathcal{L}_k(a)\,, \quad \mathcal{L}_l = \sum_a \mathcal{L}_l(a) \quad \Longrightarrow \quad [\mathcal{L}_k|\mathcal{L}_l] = \sum_{a \lesssim b} \tfrac{1}{2}\{\mathcal{L}_k(a), \mathcal{L}_l(b)\}\,.$$



Bilocal Commutator

In general, the commutator of a local charge operator $\mathcal{Q}_r$ with a bilocal operator $[\mathcal{L}_k|\mathcal{L}_l]$ again is bilocal. However, bilocal operators composed of the charges $[\mathcal{Q}_t|\mathcal{Q}_u]$

$$\frac{\mathrm{d}}{\mathrm{d}\lambda}\,\mathcal{Q}_r(\lambda) = i\Big[[\mathcal{Q}_t(\lambda)|\mathcal{Q}_u(\lambda)], \mathcal{Q}_r(\lambda)\Big],$$

yield again local operators $\mathcal{Q}_r(\lambda)$ as required.
Again, this is due to the fact that the charges $\mathcal{Q}_r$ commute:



Bilocal Commutator: Implementation

Bilocal operators $[\mathcal{L}_t|\mathcal{L}_u]$ are represented in Mathematica as

$$\texttt{Bi}[a_1\texttt{Perm}[\ldots]+\ldots+a_{l_t}\texttt{Perm}[\ldots],b_1\texttt{Perm}[\ldots]+\ldots+b_{l_u}\texttt{Perm}[\ldots]]$$

Overview of the bilocal commutator method in Mathematica:

```
CommuteBiP[BI_, P_] := (BI // DistributeBi) /.
  BI0_Bi :> (P /. P0_Perm -> CommuteBiP12[BI0, P0])

DistributeBi[B_] := B //. {
  Bi[0, x___] -> 0,
  Bi[x___, 0] -> 0,
  Bi[x__, a_ y_] -> a Bi[x, y],
  Bi[a_ x_, y__] -> a Bi[x, y],
  Bi[-Perm[x__], Y__] -> -Bi[Perm[x], Y],
  Bi[Y__, -Perm[x__]] -> -Bi[Y, Perm[x]],
  B0_Bi :> Distribute[B0]
 }
```

Bilocal Commutator: Implementation

Bilocal operators $[\mathcal{L}_t|\mathcal{L}_u]$ are represented in Mathematica as

$$\texttt{Bi}[a_1\texttt{Perm}[\ldots]+\ldots+a_{l_t}\texttt{Perm}[\ldots],b_1\texttt{Perm}[\ldots]+\ldots+b_{l_u}\texttt{Perm}[\ldots]]$$

Overview of the bilocal commutator method in Mathematica:

```
CommuteBiP[BI_, P_] := (BI // DistributeBi) /.
  BI0_Bi :> (P /. P0_Perm -> CommuteBiP12[BI0, P0])

CommuteBiP12[BI_Bi, P_Perm] := Plus[
  Bi[CommutePerm12[BI[[1]], P] , BI[[2]]] // DistributeBi,
  Bi[ BI[[1]], CommutePerm12[BI[[2]], P] ] // DistributeBi,
  CommuteBiP12loc[BI, P]
  ]

CommuteBiP12loc[BI_Bi, P_Perm] := Module[
  {LBI1 = Length[BI[[1]]], LBI2 = Length[BI[[2]]], LP = Length[P]},
  Sum[
    Module[{LongPerm = CombinePerm12[BI[[1]], BI[[2]], -d]},
      Sum[CombinePerm12[LongPerm, P, LBI1 + LBI2 + d + LP - s] -
          CombinePerm12[P, LongPerm, s],
        {s, LBI1 + d + 1, LP + LBI1 - 1}]],
    {d, 0, LP - 2}]]
```

Bilocal Commutator: Implementation

Bilocal operators $[\mathcal{L}_t|\mathcal{L}_u]$ are represented in Mathematica as

$$\texttt{Bi}[a_1\texttt{Perm}[\ldots]+\ldots+a_{l_t}\texttt{Perm}[\ldots],b_1\texttt{Perm}[\ldots]+\ldots+b_{l_u}\texttt{Perm}[\ldots]]$$

Overview of the bilocal commutator method in Mathematica:

```
CommuteBiP12loc[BI_Bi, P_Perm] := Module[
  {LBI1 = Length[BI[[1]]], LBI2 = Length[BI[[2]]], LP = Length[P]},
  Sum[
    Module[{LongPerm = CombinePerm12[BI[[1]], BI[[2]], -d]},
      Sum[CombinePerm12[LongPerm, P, LBI1 + LBI2 + d + LP - s] -
          CombinePerm12[P, LongPerm, s],
        {s, LBI1 + d + 1, LP + LBI1 - 1}]],
    {d, 0, LP - 2}]]
```



Generating Integrable Charges

We have found two types of operators that generate deformations of integrable charges

- Boost operators $\mathcal{B}[\mathcal{Q}_k]$,
- Bilocal operators $[\mathcal{Q}_t|\mathcal{Q}_u]$.

The charges $\mathcal{Q}_r$ can be deformed independently by each operator:

$$\mathcal{Q}_r = \mathcal{Q}_r(\alpha_3, \alpha_5, \ldots; \beta_{2,3}, \beta_{2,4}, \ldots, \beta_{3,4}, \beta_{3,5}, \ldots)\,,$$
$$\frac{\mathrm{d}}{\mathrm{d}\alpha_k}\,\mathcal{Q}_r = i\Big[\mathcal{B}[\mathcal{Q}_k], \mathcal{Q}_r\Big]\,, \qquad \frac{\mathrm{d}}{\mathrm{d}\beta_{t,u}}\,\mathcal{Q}_r = i\Big[[\mathcal{Q}_t|\mathcal{Q}_u], \mathcal{Q}_r\Big]\,.$$

- The set of deformations exhausts all non-trivial degrees of freedom previously obtained by brute force. [Beisert Klose]
- Specific one-dimensional deformations $\mathcal{Q}_r(\lambda)$ can be chosen by suitably defining functions $\alpha_k(\lambda)$, $\beta_{t,u}(\lambda)$.
- For the $\mathfrak{gl}(n)$ chain, a there exists a choice $\alpha_k(\lambda)$, $\beta_{t,u}(\lambda)$ that reproduces the dilatation generator (anomalous dimensions) for the $\mathfrak{su}(2)$ subsector of $\mathcal{N} = 4$ SYM.

Bethe Equations

The Bethe equations for a general symmetry group $\mathfrak{g}$ of rank R for a chain of length L are given by

$$\left(\frac{u_{a,k}+\frac{i}{2}t}{u_{a,k}-\frac{i}{2}t}\right)^L = \prod_{\substack{b=1 \\ (b,j)\neq(a,k)}}^{R}\prod_{j=1}^{M_b} \frac{u_{a,k}-u_{b,j}-\frac{i}{2}C_{a,b}}{u_{a,k}-u_{b,j}+\frac{i}{2}C_{a,b}}\,.$$

where

- $u_{a,k}$ is the rapidity of the k'th particle of type a and
- $S_{a,b}$ is the two-particle scattering matrix for particles of type a, b.

Bethe Equations

The Bethe equations for a general symmetry group $\mathfrak{g}$ of rank R for a chain of length L are given by

$$\left(\frac{u_{a,k}+\frac{i}{2}t}{u_{a,k}-\frac{i}{2}t}\right)^L = \prod_{\substack{b=1 \\ (b,j)\neq(a,k)}}^{R}\prod_{j=1}^{M_b} \frac{u_{a,k}-u_{b,j}-\frac{i}{2}C_{a,b}}{u_{a,k}-u_{b,j}+\frac{i}{2}C_{a,b}}\,.$$

boost def.

$$\boxed{\left(\frac{x(u+\frac{i}{2}t)}{x(u-\frac{i}{2}t)}\right)^L}$$

where

- $u_{a,k}$ is the rapidity of the k'th particle of type a and
- $S_{a,b}$ is the two-particle scattering matrix for particles of type a, b.

The boost deformation parameters α_k enter the *rapidity map* $x(u)$,

$$u = x + \sum_{k=3}^{\infty} \frac{\alpha_k}{x^{k-2}}\,,$$

Bethe Equations

The Bethe equations for a general symmetry group $\mathfrak{g}$ of rank R for a chain of length L are given by

$$\left(\frac{u_{a,k}+\frac{i}{2}t}{u_{a,k}-\frac{i}{2}t}\right)^{L} = \prod_{\substack{b=1 \\ (b,j)\neq(a,k)}}^{R}\prod_{j=1}^{M_b} \frac{u_{a,k}-u_{b,j}-\frac{i}{2}C_{a,b}}{u_{a,k}-u_{b,j}+\frac{i}{2}C_{a,b}}\,.$$

boost def.

$$\boxed{\left(\frac{x(u+\frac{i}{2}t)}{x(u-\frac{i}{2}t)}\right)^{L}}$$

bilocal def.

$$\boxed{e^{-2i\theta}\,\frac{u_{a,k}-u_{b,j}-\frac{i}{2}C_{a,b}}{u_{a,k}-u_{b,j}+\frac{i}{2}C_{a,b}}}$$

where

- $u_{a,k}$ is the rapidity of the k'th particle of type a and
- $S_{a,b}$ is the two-particle scattering matrix for particles of type a, b.

The boost deformation parameters α_k enter the *rapidity map* $x(u)$,

$$u = x + \sum_{k=3}^{\infty}\frac{\alpha_k}{x^{k-2}}\,,$$

while the bilocal deformations $\beta_{t,u}$ give rise to the *dressing phase*,

$$\theta = \sum_{u>t=2}^{\infty}\beta_{t,u}\Big(q_t(u)q_u(u') - q_u(u)q_t(u')\Big)\,.$$

Generating Integrable $\mathfrak{gl}(n)$ Charges: Example

Deform the charges with the boost operator $\mathcal{B}[\mathcal{Q}_3]$, i.e.

$$\frac{\mathrm{d}}{\mathrm{d}\alpha_3}\,\mathcal{Q}_r = i\big[\mathcal{B}[\mathcal{Q}_3], \mathcal{Q}_r\big],$$

To first order in α_3, the charges expand to

$$\begin{aligned}\mathcal{Q}_r &= \mathcal{Q}_r^{(0)} + \alpha_3\mathcal{Q}_r^{(3)} + \mathcal{O}(\alpha_3^2)\\ &= \mathcal{Q}_r^{(0)} + \alpha_3 i\big[\mathcal{B}[\mathcal{Q}_3^{(0)}], \mathcal{Q}_r^{(0)}\big] + \mathcal{O}(\alpha_3^2)\,.\end{aligned}$$

As a starting point, take the known short-range charges $\mathcal{Q}_r^{(0)}$.

```
Q[2, 0] = Perm[1] - Perm[2, 1] ,
Q[3, 0] = - 1/2 I (Perm[2, 3, 1] - Perm[3, 1, 2]) , . . . .
```

The first perturbative order $\mathcal{Q}_r^{(3)}$ of the first two charges read

```
In[43]:=  Q[2, 3] = I CommutePermB[B[3, 0], Q[2, 0]] // IdentifyBoundaryTerms
Out[43]=  - Perm[1] + Perm[2, 1] - 1/2 Perm[2, 3, 4, 1] + 1/2 Perm[2, 4, 1, 3]
          + 1/2 Perm[3, 1, 4, 2] - 1/2 Perm[4, 1, 2, 3]

In[44]:=  Q[3, 3] = I CommutePermB[B[3, 0], Q[3, 0]] // IdentifyBoundaryTerms
Out[44]=  1/4 I Perm[2, 4, 3, 1] + 1/4 I Perm[3, 2, 4, 1] - 1/4 I Perm[4, 1, 3, 2]
          - 1/4 I Perm[4, 2, 1, 3] - 1/2 I Perm[2, 3, 4, 5, 1] + 1/2 I Perm[2, 3, 5, 1, 4]
          + 1/2 I Perm[2, 4, 1, 5, 3] - 1/2 I Perm[2, 5, 1, 3, 4] + 1/2 I Perm[3, 1, 4, 5, 2]
          - 1/2 I Perm[3, 1, 5, 2, 4] - 1/2 I Perm[4, 1, 2, 5, 3] + 1/2 I Perm[5, 1, 2, 3, 4]
```

Generating Integrable $\mathfrak{gl}(n)$ Charges: Example

Deform the charges with the boost operator $\mathcal{B}[\mathcal{Q}_3]$, i.e.

$$\frac{\mathrm{d}}{\mathrm{d}\alpha_3}\,\mathcal{Q}_r = i\big[\mathcal{B}[\mathcal{Q}_3],\mathcal{Q}_r\big],$$

To first order in α_3, the charges expand to

$$\begin{aligned}\mathcal{Q}_r &= \mathcal{Q}_r^{(0)} + \alpha_3\,\mathcal{Q}_r^{(3)} + \mathcal{O}(\alpha_3^2)\\ &= \mathcal{Q}_r^{(0)} + \alpha_3 i\big[\mathcal{B}[\mathcal{Q}_3^{(0)}],\mathcal{Q}_r^{(0)}\big] + \mathcal{O}(\alpha_3^2)\,.\end{aligned}$$

As a starting point, take the known short-range charges $\mathcal{Q}_r^{(0)}$.
The first perturbative order $\mathcal{Q}_r^{(3)}$ of the first two charges read

```
In[43]:=  Q[2, 3] = I CommutePermB[B[3, 0], Q[2, 0]] // IdentifyBoundaryTerms
Out[43]=  - Perm[1] + Perm[2, 1] - 1/2 Perm[2, 3, 4, 1] + 1/2 Perm[2, 4, 1, 3]
          + 1/2 Perm[3, 1, 4, 2] - 1/2 Perm[4, 1, 2, 3]

In[44]:=  Q[3, 3] = I CommutePermB[B[3, 0], Q[3, 0]] // IdentifyBoundaryTerms
Out[44]=  1/4 I Perm[2, 4, 3, 1] + 1/4 I Perm[3, 2, 4, 1] - 1/4 I Perm[4, 1, 3, 2]
          - 1/4 I Perm[4, 2, 1, 3] - 1/2 I Perm[2, 3, 4, 5, 1] + 1/2 I Perm[2, 3, 5, 1, 4]
          + 1/2 I Perm[2, 4, 1, 5, 3] - 1/2 I Perm[2, 5, 1, 3, 4] + 1/2 I Perm[3, 1, 4, 5, 2]
          - 1/2 I Perm[3, 1, 5, 2, 4] - 1/2 I Perm[4, 1, 2, 5, 3] + 1/2 I Perm[5, 1, 2, 3, 4]
```

We can verify that the deformed charges indeed commute:

```
In[45]:= CommutePerm[Q[2, 3], Q[3, 0]] + CommutePerm[Q[2, 0], Q[3, 3]]
Out[45]= 0
```

Summary

- Perturbative long-range integrable spin chains can be obtained as deformations of short-range models via a generating equation.
- Suitable deformation operators are given by boosts $\mathcal{B}[\mathcal{Q}_k]$ and bilocal operators $[\mathcal{Q}_t|\mathcal{Q}_u]$ constructed from the integrable charges $\mathcal{Q}_r$.
- The deformation reproduces all degrees of freedom that were obtained before by brute force.
- The deformations give rise to the rapidity map $x(u)$ (boosts) and the dressing phase θ (bilocal operators).

For generic symmetry algebra $\mathfrak{g}$, *all charges* are defined to *all orders* (on an infinite chain) and are integrable by construction.